Center for Night Vision and Electro-Optics

AMSEL-NV-TR-0057

HP-41CX PROGRAMS FOR HgCdTe DETECTORS AND IR SYSTEMS

by
DAVID R. KAPLAN

October 1987

FORT BELVOIR, VIRGINIA 22060-5677

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE

A188 935

REPORT DOCUMENTATION PAGE

Form Approved
OMB No 0704 0188
Exp Date Jun 30 1986

1a REPORT SECURITY CLASSIFICATION: UNCLASSIFIED

1b RESTRICTIVE MARKINGS:

2a SECURITY CLASSIFICATION AUTHORITY:

2b DECLASSIFICATION / DOWNGRADING SCHEDULE:

3 DISTRIBUTION / AVAILABILITY OF REPORT: Approved for public release; distribution unlimited.

4 PERFORMING ORGANIZATION REPORT NUMBER(S): Technical Report AMSEL-NV-TR-0057

5 MONITORING ORGANIZATION REPORT NUMBER(S):

6a NAME OF PERFORMING ORGANIZATION: Center for Night Vision and Electro-Optics (CNVEO)

6b OFFICE SYMBOL *(If applicable)*: AMSEL-RD-NV-IT

7a NAME OF MONITORING ORGANIZATION:

6c. ADDRESS *(City, State, and ZIP Code)*: Fort Belvoir, VA 22060-5677

7b ADDRESS *(City, State, and ZIP Code)*:

8a. NAME OF FUNDING / SPONSORING ORGANIZATION: CNVEO

8b OFFICE SYMBOL *(If applicable)*: AMSEL-RD-NV-IT

9 PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER:

8c. ADDRESS *(City, State, and ZIP Code)*: Fort Belvoir, VA 22060-5677

10 SOURCE OF FUNDING NUMBERS

PROGRAM ELEMENT NO	PROJECT NO	TASK NO	WORK UNIT ACCESSION NO

11 TITLE *(Include Security Classification)*: HP-41CX PROGRAMS FOR HgCdTe DETECTORS AND IR SYSTEMS

12 PERSONAL AUTHOR(S): David R. Kaplan

13a TYPE OF REPORT: Final

13b TIME COVERED: FROM ________ TO ________

14 DATE OF REPORT *(Year, Month, Day)*: October 1987

15 PAGE COUNT: 42

16. SUPPLEMENTARY NOTATION:

17 COSATI CODES

FIELD	GROUP	SUB-GROUP

18 SUBJECT TERMS *(Continue on reverse if necessary and identify by block number)*

Pocket Computer	HgCdTe	Photo Sensor
Programs	Detectors	Analysis
HP-41	Infrared	IR Systems

19 ABSTRACT *(Continue on reverse if necessary and identify by block number)*

Programs have been written for the HP-41CX pocket computer which aid in the analysis of HgCdTe detectors, focal planes, and infrared systems. They have been written to run in a basic HP-41CV or HP-41CX with no card reader or additional ROMs required. An HP-41C with less than maximum RAM will be unable to hold all the programs in memory at the same time, but a subset can be loaded by hand from the enclosed listings. The purpose of this report is to describe the programs and to provide the user with information needed to use them.

20 DISTRIBUTION / AVAILABILITY OF ABSTRACT
☒ UNCLASSIFIED/UNLIMITED ☐ SAME AS RPT ☐ DTIC USERS

21 ABSTRACT SECURITY CLASSIFICATION: UNCLASSIFIED

22a NAME OF RESPONSIBLE INDIVIDUAL: David R. Kaplan

22b TELEPHONE *(Include Area Code)*: (703) 664-3063

22c OFFICE SYMBOL: AMSEL-RD-NV-IT

DD FORM 1473, 84 MAR

83 APR edition may be used until exhausted
All other editions are obsolete

SECURITY CLASSIFICATION OF THIS PAGE
UNCLASSIFIED

CONTENTS

HP-41CX PROGRAMS
FOR
HgCdTe DETECTORS AND IR SYSTEMS

I. INTRODUCTION

Programs have been written for the HP-41CX which aid in the analysis of HgCdTe detectors, focal planes, and infrared systems. They have been installed as a "bundle" in an HP-41CX and a "WALL" (write-all) operation has been performed, saving the group of programs and machine status to magnetic cards. They have been written to run in a basic HP-41CV or HP-41CX with no card reader or additional ROMs required. An HP-41C with less than maximum RAM will be unable to read this WALL configuration and will be unable to hold all the programs in memory at the same time, but a subset can be loaded by hand from the enclosed listings.

The purpose of this report is to describe the status of the HP-41CX after the WALL cards have been read, and to provide the user with information needed to use the programs.

It is assumed that the user has some familiarity with the operation and programming of the HP-41 family of pocket calculators. However, the description will begin with a brief review of the more pertinent characteristics and procedures.

The HP-41CX can have many different unique programs in memory at the same time. Jumps to perform a subroutine or to alter program flow are referred to labels (LBLs) which identify the target location. There are a limited number of short (quick) labels available, and when writing a program, it's important to keep track of label usage. The HP-41CX provides the capability, when desired, of separating programs with an END statement. The short labels cannot be reached by GTO (go to, jump) or XEQ (execute, gosub) commands beyond an END statement. This means that such labels are local labels and don't conflict with identical labels in other programs separated by an END statement. There are, however, global labels available which permit the user to go directly to a program from anywhere in RAM. These also permit one program to GTO or XEQ another program, even though they are separated by an END statement. Performing a CATALOG 1 causes the HP-41CX to list the global labels. The listing may be paused with the R/S key, and you may step forward or backward through the listing with the SST and BST keys. You may put yourself in a desired program by listing CATALOG 1, pausing with the desired program in the display, and pressing the ← key.

The global labels of the programs included in this WALL configuration are listed on the next page. Global labels need not necessarily be separated by END statements, and some of the labels in this group of programs, in fact, are not. Indented programs are included in the same local area with program above.

LABEL	DESCRIPTION
FLAG	Puts calculator flags back to a default status.
NEdt	Noise equivalent delta temperature of an IR system, and photons collected in an integration time.
DET RNG	Estimates detection range of a target.
D*	D-Star (PV, Johnson, and BLIP), RoA BLIP, D*lambda to D*BB conversion factor.
QBB	Blackbody photon flux.
BB	General purpose blackbody calculations.
SV	General purpose root finder used by other programs.
GR-DIFF	RoA, temperature, cutoff interpolator/extrapolator.
HCT	X of Hg(x)Cd(1-x)Te from cutoff and temperature.
EG	Bandgap from x and temperature.
NI	Intrinsic carrier concentration from material parameters.
RØA	RoA product from material parameters.
CONSTNT	Useful physical constants.
INJEFF	Direct Injection injection efficiency.

The following USER key definitions are also included. (In USER mode, the following keys perform the indicated operation.)

PRESSING	WILL RESULT IN
TAN	PACK
SHIFT TAN	CLRG (clear all data registers)
SHIFT R/S	Run the FLAG program

Twenty-six registers have been reserved for data storage (registers 00 through 25.) Thirty-five registers of the 319 available in an empty machine are unused and available for either additional programming or additional data storage. Each register has room for approximately five program steps. Should more room be needed for a different, lengthy program, and if you are using an HP-41CX or an HP-41C(V) with an extended memory module, you may SAVEP any program in RAM to EXTENDED MEMORY (which is similar to a RAM-DISK) and then clear the program from RAM. This may be repeated as necessary with other programs until EXTENDED MEMORY is full. Programs in EXTENDED MEMORY cannot be run, but you may GETP them back into RAM, if there is enough RAM available.

Of the 26 data registers available, only the first 15 STO (store) and RCL (recall) statements are quick access (one-byte statements instead of two-byte statements.) You can ignore this fact, but it is the reason that the first 15 registers are used most often. Since there are more than 15 different physical parameters involved in the above programs and some data registers are needed for bookkeeping or temporary scratch storage, there is not a firm relationship from global program to global program between any physical parameter and any data register. You can rest assured that the right data are in the right place at the right time when running a given program. BUT don't assume that data entered while running one program will be correctly placed, should you go to another global program which requires the same data. It is advisable to perform a CLRG (SHIFT-TAN in USER mode) and reenter all necessary data when moving to a new program.

There is a similar caution concerning the HP-41CX flags. There are 56 flags which are used by the HP-41CX, 30 of which are user alterable. The status of the first five are indicated on the LCD display. Since the calculator is slow (as computers go), some calculations are done once, the data are stored, and a flag is set to let the program know that the calculation need not be done again, unless certain parameters are altered. If these parameters are altered, the flag is automatically cleared so that a new calculation will be performed when required. Flags are used to control program flow for other reasons as well. To keep the user informed, only the first five flags (which are visible in the display) are used in this fashion. Different global programs may use the same flag to mean different things, so it is advisable when moving to a new global program (or just starting a program) to check to see that the five displayed flags (0 through 4) are clear. If you can't see them, they're clear. Running the FLAG program (SHIFT-R/S in USER mode) will insure that all flags are clear. This will leave you in the FLAG program, and then you will have to go to the program you wish to run.

Running a program generally proceeds as follows. With the exception of FLAG which may be run at any time by pressing SHIFT R/S, the user must, first, position himself within the program of interest. This may be done, as described above, by using CATALOG 1 and pressing R/S to stop the catalog scan with the desired program name in the display, or by going directly to the program of interest with the GTO command. For example, to run the blackbody program (BB), press keys as follows: SHIFT GTO ALPHA BB ALPHA.

Make sure that flags 0 through 4 are clear. It may be wise to do a CLRG to increase the likelihood that, should you fail to provide all the necessary data to the program, an error will occur or invalid results will be clearly invalid.

The programs generally require data to be input before any results may be computed. The top row of keys (A, B, C, D, E) and their SHIFTed counterparts (a, b, c, d, e) are used to communicate with a program. This is because, *WHEN IN USER MODE,* pressing one of these keys will begin program execution at the label (LBL) which matches the key. These labels are local. If no such label exists in the current program area, then the function identified on the keyboard is executed instead. For example, suppose the current program area contains the instruction sequence:

```
.
.
.
LBL d
5
*
1
+
RTN
.
.
.
```

With USER mode active, pressing SHIFT D will cause the program to being running at LBL d, where it will multiply the contents of the X register (the display) by 5, then add 1 and, finally, stop showing the answer again in the X register.

The programs are written so that pressing certain keys will cause data which have been entered into the X (or X and Y) (or X, Y, and Z) register to be stored away for later use. Pressing certain keys (perhaps after some additional data entry) will cause desired calculations to be performed and the results displayed. Which keys do what varies from program to program, and the functions are not indicated on the keyboard. You must, therefore, either remember the key definitions or refer to a document such as this report. It is common, when an individual, dedicated program is stored to a magnetic card, to write the definitions for each key on that magnetic card. The card is conceptually divided into five parts horizontally representing the five keys in the top row, and two parts vertically representing the unshifted and shifted labels. The upper half represents the shifted keys. Although this has not been done with the WALL cards, the same convention will be followed by including in each program description a drawing based on this layout which defines the operation of each key.

It is somewhat inconvenient to use a single key to enter multiple data, but it is necessary, since there is often more data required for a calculation than there are keys. Referring to one of these drawings, ENTER means press the enter key on the keyboard after keying in the requested data. This pushes the data just entered up into the Y register and permits the user to enter additional data in the X register. Another ENTER pushes the contents of Y into Z, X into Y, so additional data may be placed in X. After keying in all the requested data for a key, press that key so that the program will store it. Continue from key to key, entering in all data required to compute the parameter desired. Note that some programs calculate more than one parameter and that some parameters may not require that all the data indicated on the drawing be input. Which parameters require which data comes from experience or referral to this report.

Once all the necessary data have been keyed and stored, pressing the FIND key for the desired parameter will initiate the calculation to produce the answer for that parameter. Obviously, pressing FIND before all the necessary support data have been entered will, very likely, produce invalid results. It may even produce an operational error; e.g., divide by zero, causing the calculation to terminate. Under most circumstances no harm is done when this occurs. However, *NOTE* that some programs call other programs. If the error occurs in the called program, you will be left in that program, not in the program from which you started. It is advisable, should such an improper termination occur, to re-position yourself via CATALOG 1 or GTO to the program of interest. It is probably, also, wise to clear any set flags.

If incorrect data have been keyed and stored, it may be corrected by performing the correct data entry sequence for the key dedicated to that data. This means that you will have to reenter all the data associated with that key. This is also true if you wish to change any data, either before or after FINDing a result. To recalculate a result with different initial conditions, you need only reenter the data associated with the key whose data you wish to alter. Then, press the appropriate FIND key for the result desired.

What follows is user documentation and description for each of the programs.

II. PROGRAM DOCUMENTATION

1. FLAG. This is a very short program designed specifically to restore all 55 flags of the HP-41CX to default status. It clears flags 0 through 25, sets the display to FIX 2 format, sets DEG mode, calls for a period as the decimal point, and calls for commas between digit groupings of three. It alters no data registers. This program may be quickly run by pressing SHIFT R/S in the USER mode if FLAG has been assigned to this key.

The FLAG program on the WALL cards contains undocumented program steps and should not be altered or single-stepped in RUN mode.

Also, included is a listing for FLAG which simulates the short version and which contains no undocumented operations. This version is significantly slower than the short version, but it may be keyed by hand and needs no precautions.

PROGRAM LISTINGS FOR FLAG

Two listings are given. The first is very fast and short but contains undocumented code which can only be created by procedures unavailable to most users. It is included on the WALL cards. The second is longer and significantly slower but may be easily keyed by hand.

```
01/10/1986

01♦LBL "FLAG"
02 "<x̄♦"
03 ASTO d
04 CF 03
05 .END.
```

```
01♦LBL "FLAG"
02 .025
03♦LBL 00
04 CF IND X
05 ISG X
06 GTO 00
07 SF 26
08 SF 27
09 SF 28
10 SF 29
11 FIX 2
12 DEG
13 RTN
14 END
```

2. NEdT, DET RNG.



Below is a table showing the data required, registers used, sample test data, and results.

	REQ'D FOR NEdT	REQ'D FOR PHOTONS	REQ'D FOR DET RNG	REGISTER USED	TEST DATA
LAMBDA INITIAL	X	X	X	1	3
LAMBDA FINAL	X	X	X	10	5
F/# COLD SHIELD	X		X	17*	1
F/# SIGNAL	X	X	X	4	1.4
BACKGROUND TEMPERATURE	X	X	X	3	300
DETECTOR TEMPERATURE	X		X	14	200
INTEGRATION TIME	X	X	X	5	4E-5
QUANTUM EFFICIENCY	X	X	X	18	60
DETECTOR AREA	X	X	X	2	1E-5
RoA	X		X	15	50
TARGET DELTA TEMP			X	16	3
TARGET AREA			X	11	3
SYSTEM FOCAL LENGTH			X	12	100
S/N CRITERIA			X	13**	10
ATM. TRANSMISSION (PERCENT/KM)			X	20***	70.71

RESULTING NEdT = 0.15511 K
RESULTING SIGNAL PHOTONS = 4.02 E5
RESULTING DETECTION RANGE = 3.885 km

Other registeres used:
0,6,7,8,9

* This register actually contains the ratio of the f/numbers squared.
** This register used for scratch as well, so S/N not saved.
*** SQR (transmission percent/100) actually stored.

This program finds (1) Noise equivalent delta temperature (NEdT), (2) number of photons collected by a detector (or an equivalent detector, for TDI) through the system in an integration time, and (3) detection range for a target.

The NEdT for an IR system is calculated from certain system and detector parameters. It is a relatively abbreviated calculation, so the results are only approximate. Assumptions made include that the noise comes only from the detector or the background, that the noise bandwidth is the reciprocal of the integration time, that the quantum efficiency is constant with respect to wavelength, and that the detector noise is due to shot noise at zero bias.

The calculation performed is:

$$\text{NEdT} = \frac{\text{SQR}(2*\text{PC} + (4kTt/\text{Ro}/e\wedge 2))}{\text{SC}},$$

where Q.E. is the detector quantum efficiency;

t = detector integration time;

T = detector temperature;

Ro = detector zero bias dynamic resistance;

A = detector active area; and

PC is the charge absorbed by the detector/t due to photons.

or,

$$\text{PC} = \frac{Q\ A\ (\text{Q.E.})\ t}{(2 * (\text{cold shield F/\#}))\wedge 2},$$

where Q = photon flux rate from cuton to cutoff/cm²/s, and

SC is the signal charge absorbed by the detector/t per degree change in scene temperature.

or,

$$\text{SC} = \frac{dQ}{dT}\ \frac{A\ (\text{Q.E.})\ t}{(2 * (\text{signal F/\#}))\wedge 2},$$

where $\frac{dQ}{dT}$ = change in signal flux per change in scene temp.

The photons per integration time, found by pressing key B, is the number of SIGNAL photons collected through the optics of the indicated system (using the signal F/#) which produces signal carriers. For TDI systems the integration time entered should be the integration time per detector times the number of detectors in TDI, so that the number of carriers accumulated is the number for the TDI column. To find the number of background carriers collected, set the signal F/# equal to the background F/#.

The detection range is calculated by finding the range at which the image of the target is sufficiently smaller than the detector, so that the S/N is as desired. The signal is attenuated, based upon the entered atmospheric transmission per kilometer. The SV (solve) program is called to perform the iteration to a solution.

$$\text{S/N} = (\text{transmission})\wedge(\text{range}) \times \frac{(\text{target delta T})}{\text{NEdT}} \times \frac{(\text{target area})}{A} \times \frac{f\wedge 2}{\text{Range}\wedge 2}$$

PROGRAM LISTING FOR NEdT, DET RNG

```
01/10/1986

01◆LBL "DET RNG"
02◆LBL C
03 STO 12
04 RDN
05 STO 11
06 RDN
07 STO 16
08 RTN
09◆LBL D
10 100
11 /
12 SQRT
13 STO 20
14 X<>Y
15 STO 13
16 XEQ "NEdT"
17 RCL 13
18 *
19 RCL 16
20 X<>Y
21 "NO DETECT"
22 X>Y?
23 PROMPT
24 /
25 RCL 12
26 X↑2
27 RCL 11
28 *
29 RCL 02
30 /
31 *
32 SQRT
33 1 E4
34 /
35 STO 13
36 SF 10
37 "R"
38 ASTO 06
39 0
40 X<>Y
41 FIX 3
42 XEQ "SV"
43 CLA
44 ARCL X
45 "├KM"
46 PROMPT
47◆LBL "R"
48 RCL 20
49 X<>Y
50 Y↑X
51 RCL 13
52 *
53 RCL 07
54 -
55 RTN
56◆LBL "NEdT"
57◆LBL A
58 .5
59 ST+ 03
60 SF 00
61 XEQ 08
62 STO 06
63 1
64 ST- 03
65 XEQ 08
66 CF 00
67 .5
68 ST+ 03
69 RDN
70 RCL 06
71 X<>Y
72 ST- 06
73 +
74 RCL 17
75 /
76 215 E13
77 RCL 14
78 *
79 RCL 15
80 /
81 RCL 02
82 *
83 RCL 05
84 *
85 +
86 SQRT
87 RCL 06
88 /
89 FIX 5
90 "NEdT="
91 ARCL X
92 AVIEW
93 RTN
94◆LBL a
95 STO 10
96 X<>Y
97 STO 01
98 RTN
99◆LBL b
100 X<Y?
101 GTO 99
102 STO 04
103 /
104 X↑2
105 STO 17
106 RTN
107◆LBL c
108 STO 14
109 X<>Y
110 STO 03
111 RTN
112◆LBL d
113 STO 18
114 X<>Y
115 STO 05
116 RTN
117◆LBL e
118 STO 02
119 RTN
120◆LBL E
121 STO 15
122 RTN
123◆LBL B
124 CF 00
125◆LBL 08
126 SF 01
127 RCL 01
128 XEQ 01
129 STO 00
130 RCL 10
131◆LBL 01
132 14388
133 X<>Y
134 /
135 RCL 03
136 /
137 STO 08
138 2
139 +
140 RCL 08
141 *
142 2
143 +
144 RCL 08
145 E↑X
146 STO 07
147 /
148 RCL 08
149 X↑2
150 RCL 03
151 +
152 .5
153 +
154 2
155 /
156 RCL 07
157 X↑2
158 /
159 +
160 RCL 08
161 9
162 *
163 6
164 +
165 RCL 08
166 *
167 2
168 +
169 27
170 /
171 RCL 07
172 X↑2
173 /
174 RCL 07
175 /
176 +
177 RCL 03
178 X↑2
179 *
180 RCL 03
181 *
182 1581 E5
183 *
184 RCL 18
185 *
186 RCL 04
187 X↑2
188 /
189 RCL 02
190 *
191 RCL 05
192 *
193 FS?C 01
194 RTN
195 RCL 00
196 -
197 FS? 00
198 RTN
199 SCI 2
200 CLA
201 ARCL X
202 "├ PHOT"
203 AVIEW
204 RTN
205 END
```

3. **D*, QBB.**

Lambda Initial ENTER Lambda Final (μM)	Quant. Eff. (percent) ENTER Field of View (full angle) (degrees)	Bkgnd temp ENTER Detector temp (Kelvin)	RoA (ohm-cm^2)	FIND RoA BLIP
SHIFT A	SHIFT B	SHIFT C	SHIFT D	SHIFT E
FIND D* (PV)	FIND D* (BLIP)	FIND D* (JOHNSON)	source temp FIND D*lambda to D*BB conversion factor	FIND photon flux with Q.E. and FOV
A	B	C	D	E

Below is a table showing the data required, registers used, sample test data, and results.

	REQ'D FOR D*PV	REQ'D FOR D*BLIP	REQ'D FOR D*JHNSN	REQ'D FOR C.F.	REQ'D FOR P. FLUX	REQ'D FOR RoABL	RGSTR USED	TEST DATA
LAMBDA INITIAL	X	X		X	X	X	5	8
LAMBDA FINAL	X	X		X	X	X	10	11
QUANTUM EFFICIENCY	X	X			X	X	11	50
FIELD OF VIEW	X	X			X	X	13	120
BACKGROUND TEMP.	X	X				X	12	300
DETECTOR TEMPERATURE	X		X			X	14	80
RoA	X		X				9	0.1
SOURCE TEMPERATURE				X			12*	500

RESULTING D*PV = 1.93 E10
RESULTING D*BLIP = 4.82 E10
RESULTING D*JOHNSON = 2.11 E10
RESULTING CONVERSION FACTOR = 5.60
RESULTING PHOTON FLUX = 1.65 E17/CM^2/SEC
RESULTING RoA BLIP = 0.522

OTHER REGISTERS USED:
0,6,7,8,15

*Source temperature is placed temporarily in register 12; background temperature is restored to register 12, at the end of the calculation.

This program contains routines to find six different parameters associated with photovoltaic detector D-star.

(1) D*JOHNSON is the detectivity based on the assumption that the detector shot noise at zero bias is the only noise source. It is computed from:

$$D^*JOHNSON = \frac{e\,(Q.E.)\,(cutoff)\,SQR(Ro^*A/(4kT))}{hc}.$$

(2) D*BLIP is the detectivity based on the assumption that the background noise is the only noise source. It is computed from:

$$D^*BLIP = \frac{(cutoff)\,SQR((Q.E.)/(photon\ flux))}{SQR(2)\ h\ c\ SIN(FOV/2)}.$$

(3) D*PV is the detectivity which includes both detector and background noise sources and is computed from the above as:

$$D^*PV = \frac{(D^*JOHNSON)\,(D^*BLIP)}{SQR((D^*JOHNSON)^\wedge 2 + (D^*BLIP)^\wedge 2)}.$$

(4) Blackbody photon flux is the photon collection rate per second per square centimeter of detector area for the given quantum efficiency and field of view. It is found by twice calling a routine which calculates the flux from lambda = 0 up to lambda. Lambda is first set to cutoff, then to cuton, and the results are subtracted. The basic routine which is called is:

$$Q = a\,\frac{SUM(X^\wedge 2 + 2^*X/n + 2/n^\wedge 2)}{n\ EXP\ (nX)}, \qquad \text{for } n = 1,2,3 \text{ and } X = \text{lambda, and}$$

where a = 2 PI c (kT/(hc))^3.

(5) The conversion factor is the factor which converts a blackbody D* to a lambda peak D*. It is a theoretical value which, here, assumes that the quantum efficiency of the detector is constant between cuton and cutoff. For this calculation, it is assumed that the peak response is at the cutoff. It is calculated from (where sigma is the Stefan-Boltzmann constant):

C.F. = (photon flux) (b/a), where a is as above, and

$$b = \frac{2\ PI\ k^\wedge 3}{c\ (sigma)\ (source\ temperature)\ h^\wedge 2\ (lambda\ cutoff)}.$$

(6) RoA BLIP is the value of the RoA for which D*JOHNSON and D*BLIP equal one another. It is, therefore, the RoA for which the D*PV is .7071 of D*BLIP. Requesting this calculation presents the answer in the display and, also, stores this value in register 9. This means that further calculations for D* will be based on this RoA.

PROGRAM LISTING FOR D*, QBB

```
01/10/1986

01♦LBL e
02 1
03 STO 09
04 XEQ B
05 XEQ C
06 /
07 X↑2
08♦LBL d
09 STO 09
10 RTN
11♦LBL a
12 CF 00
13 STO 10
14 X<>Y
15 STO 05
16 RTN
17♦LBL b
18 CF 00
19 STO 13
20 X<>Y
21 STO 11
22 RTN
23♦LBL c
24 CF 00
25 STO 14
26 X<>Y
27 STO 12
28 RTN
29♦LBL "D*"
30♦LBL A
31 XEQ B
32 XEQ C
33 R-P
34 STO 08
35 P-R
36 *
37 RCL 08
38 /
39 RTN
40♦LBL E
41 SCI 2
42 RCL 15
43 FS? 00
44 RTN
45 RCL 05
46 XEQ 08
47 STO 06
48 RCL 10
49 XEQ 08
50 RCL 06
51 -
52 FS? 02
53 RTN
54 RCL 13
55 2
56 /
57 SIN
58 X↑2
59 *
60 RCL 11
61 %
62 STO 15
63 SF 00
64 RTN
65♦LBL B
66 XEQ E
67 SQRT
68 RCL 11
69 X<>Y
70 /
71 RCL 10
72 *
73 356 E14
74 *
75 RTN
76♦LBL C
77 SCI 2
78 RCL 09
79 RCL 14
80 /
81 SQRT
82 RCL 10
83 *
84 RCL 11
85 *
86 10854 E5
87 *
88 RTN
89♦LBL "QBB"
90♦LBL 08
91 3
92 STO 07
93 CLX
94 14388
95 X<>Y
96 /
97 RCL 12
98 /
99 STO 08
100 CLX
101♦LBL 00
102 2
103 RCL 07
104 /
105 RCL 08
106 +
107 RCL 08
108 *
109 2
110 RCL 07
111 X↑2
112 /
113 +
114 RCL 07
115 /
116 RCL 08
117 RCL 07
118 *
119 E↑X
120 /
121 +
122 DSE 07
123 GTO 00
124 FS? 02
125 RTN
126 RCL 12
127 X↑2
128 RCL 12
129 *
130 6324 E7
131 *
132 *
133 RTN
134♦LBL D
135 X<> 12
136 STO 00
137 CF 00
138 SF 02
139 XEQ E
140 CF 02
141 2216
142 *
143 RCL 10
144 /
145 RCL 12
146 /
147 RCL 00
148 STO 12
149 RDN
150 1/X
151 END
```

4. **BB.**

INITIALIZE	FIND or ENTER Blackbody Temp. (K)	FIND or ENTER Cutoff (μM)	COMPARE Lambda Max versus BB Temp. (maximized photon flux)	COMPARE Lambda Max versus BB Temp. (maximized power)
SHIFT A	SHIFT B	SHIFT C	SHIFT D	SHIFT E
PREFIX	PREFIX	PREFIX	FIND	FIND
$\frac{d}{dT}$	integral to infinity wrt lambda	integral to cutoff	Photon Flux	Power Flux
		CHOICE		
A	B	C	D	E

Below is a table showing sample test data and results.

LAMBDA FINAL = 5 μM (not required for integral to infinity)
BLACKBODY TEMP. = 300K REGISTERS USED: 0 THRU 15

RESULTS

PREFIXES PRESSED	**Q (photons)**	**W (watts)**
NONE	2.058 E16 $/cm^2/s/\mu M$	8.175 E-4 $/cm^2/\mu M$
INTEGRAL TO INFINITY (ITI)	4.105 E18 $/cm^2/s$	4.592 E-2 $/cm^2$
INTEGRAL TO CUTOFF (ITC)	1.320 E16 $/cm^2/s$	5.901 E-4 $/cm^2$
DERIVATIVE WRT TEMP (DWT)	6.580 E14 $/cm^2/s/\mu M/K$	2.614 E-5 $/cm^2/\mu M/K$
DWT, ITI	4.105 E16 $/cm^2/s/K$	6.123 E-4 $/cm^2/K$
DWT, ITC	4.749 E14 $/cm^2/s/K$	2.149 E-5 $/cm^2/K$

The COMPARE functions permit you to key in either wavelength or temperature and calculate the other (temperature or wavelength) for maximum flux. For example, keying in 300 and pressing SHIFT D will produce the value 12.23. This means that for a 300 K blackbody, the wavelength of peak photon flux density is 12.23 μM. Or conversely, a blackbody whose peak photon flux density is at 300 μM wavelength has a temperature of 12.23 K. The operation for SHIFT E is the same, except the criterion is the wavelength for peak power density. Keying 300 and SHIFT E produces the value 9.659. NOTE: The data entered prior to a COMPARE operation does not alter any values that have been stored for wavelength or temperature via SHIFT B or SHIFT C.

This program uses the keys a little differently. First, before doing anything else, you must initialize the program by pressing SHIFT A. This configures the HP41 data registers and flags for the rest of the program. It also is used to clear partial entries, as described below.

Second, at most only two initial conditions are required, wavelength and temperature, and the keys which are used to store these conditions may also be use to recall (or verify) them. The HP41 has a flag which it uses to keep track of whether a numeric key (0 through 9) has been pressed. This flag is not visible in the display. As soon as any numeric key is pressed, this flag gets set. The program uses this fact to determine if the user is storing data or requesting an answer. If the flag is set, the user is assumed to be storing data, and the value in the display is stored as the new value of the variable designated by the key pressed. If the flag is clear, the present value of the requested variable is displayed. The program clears this flag automatically after every operation, so the presence (or absence) of new numeric key presses may be sensed. NOTE: If this flag is set and any of the unshifted keys are pressed, the program updates the wavelength with the current value in the X register (the display), in addition to performing the desired operation. If you have done some simple keyboard calculations (thereby setting the flag), and you now wish to perform a program operation without altering the value for wavelength, press the INITIALIZE key, first. This clears the flag without altering the current values for wavelength and temperature.

Third, keys A, B, and C are not used for data entry, but may be used for configuring the program for the type of calculation to be performed. For example, pressing D or E without first pressing A, B, or C will result in the calculation of the photon flux density in photons/cm²/s/μM or power density in W/cm²/μM, respectively. See the table showing sample test data and results. Pressing A first, for example, then pressing D or E will result, instead, in the calculation of the change in flux (either photons or power) per degree change in blackbody temperature. You may press pairs of prefix keys prior to calling for a calculation (e.g.,A, C, E to find the change in power up to the cutoff per degree Kelvin). The prefix keys may be pressed in any sequence as long as you end up pressing D or E. NOTE: It doesn't make sense to use both B and C. The last one pressed takes precedence. If you have pressed any prefix keys and change your mind (or have made an error), pressing the INITIALIZE key will cancel and give you a fresh start. Stored values for wavelength and temperature are unaffected.

At the end of a calculation, the program displays in sequence (1) the key sequence which has been performed, (2) the temperature that was used, (3) the wavelength that was used, and (4) finally, the answer. The key sequence is displayed numerically proceeding from left to right, where the numbers represent the respective unshifted keys. Key A is 1, key B is 2; etc. For example, pressing C, A, D will display a key sequence of 314.000, where the number of trailing zeros depends upon the current display setting.

The following equations and procedures are used in BB (L = wavelength).

$$\text{EQN1} = c_1(T/c_2)^{l-p}\,\text{SUM}_{n=1}^{\text{inf}}\,[(l+p-1)!/n^2\,\text{EXP}(-nc_2/LT)\,\text{SUM}_{m=0}^{l+p-1}\,\{(nc_2/LT)^m/m!)\}].$$

$$\text{EQN2} = c_1/[L^{l+p+1}\,(\text{EXP}\{c_2/LT\}-1)].$$

$$\text{EQN3} = c_2^l\,\text{EXP}\,(c_2/LT)\,/\,[T^2\,(\text{EXP}\{c_2/LT\}-1)].$$

$$\text{EQN4} = zc_1\,(T/c_2)^{l-p}\,(l+p-1)!$$

Default conditions are: $p=0$, $c_1 = hc^2/(2Pi) = 37415$, $c_2 = hc/k = 14388$, $zc_1 = 40495$ and use EQN2.

If prefix key A is used, then $p=1$; if using EQN2, then multiply by EQN3, and divide answer by c_2.

If prefix key B is used, then use EQN4, instead.

If prefix key C is used, then use EQN1, instead.

If calculating power, then $l = 4$.

If calculating photon flux, then $l = 3$; use $c_1 = 1/k = 18837E19$, instead; if using EQN4, then use $zc_1 = 22642E19$, instead.

Using these equations and logical tests in this fashion permits calculation of many different parameters in a small program space.

PROGRAM LISTING FOR BB

```
01/10/1986

01♦LBL "BB"
02♦LBL a
03 37415
04 STO 04
05 14388
06 STO 10
07 18837 E19
08 STO 03
09 CLX
10♦LBL 01
11 0
12 STO 15
13 STO 14
14 STO 09
15 X<>Y
16 CF 00
17 SF 01
18 CF 22
19 SCI 3
20 RTN
21♦LBL b
22 FS?C 22
23 STO 11
24 RCL 11
25 RTN
26♦LBL A
27 CF 01
28 XEQ 07
29 1
30 GTO 05
31♦LBL B
32 SF 02
33♦LBL C
34 XEQ 07
35 3
36 FS?C 02
37 2
38 STO 15
39♦LBL 05
40 RCL 14
41 10
42 *
43 +
44 STO 14
45 FIX 0
46 RTN
47♦LBL 07
48♦LBL c
49 SCI 3
50 FS?C 22
51 STO 12
52 RCL 12
53 RTN
54♦LBL D
55 SF 00
56♦LBL E
57 XEQ 07
58 RCL 10
59 RCL 11
60 /
61 STO 13
62 RCL 12
63 X≠0?
64 /
65 STO 08
66 4
67 FS? 00
68 3
69 STO 05
70 STO 01
71 STO 02
72 1
73 STO 00
74 FS? 01
75 GTO 14
76 ST+ 01
77 ST- 02
78♦LBL 14
79 +
80 XEQ 05
81 RCL 04
82 FS? 00
83 RCL 03
84 STO 06
85 GTO IND 15
86♦LBL 00
87 RCL 12
88 RCL 01
89 1
90 +
91 Y↑X
92 /
93 RCL 08
94 E↑X
95 STO 09
96 1
97 -
98 /
99 FS? 01
100 GTO 01
101 LASTX
102 /
103 RCL 09
104 *
105 RCL 13
106 X↑2
107 *
108 RCL 10
109 /
110 GTO 01
111♦LBL 02
112 1.202
113 ENTER↑
114 1.0823
115 FS? 00
116 X<>Y
117 STO 09
118 GTO 02
119♦LBL 03
120 RCL 01
121 STO 15
122 0
123 RCL 08
124 RCL 00
125 *
126 STO 07
127♦LBL 04
128 RCL 15
129 1
130 -
131 Y↑X
132 LASTX
133 FACT
134 /
135 +
136 RCL 07
137 DSE 15
138 GTO 04
139 E↑X
140 /
141 RCL 00
142 RCL 05
143 Y↑X
144 /
145 ST+ 09
146 RCL 09
147 /
148 FS? 00
149 R-D
150 FIX 3
151 RND
152 ISG IND 15
153 CLA
154 X≠0?
155 GTO 03
156♦LBL 02
157 RCL 06
158 RCL 13
159 RCL 02
160 Y↑X
161 /
162 RCL 01
163 1
164 -
165 FACT
166 *
167 RCL 09
168 *
169 FS? 01
170 GTO 01
171 RCL 10
172 /
173 GTO 01
174♦LBL e
175 .7896
176 /
177♦LBL d
178 3670
179 /
180 1/X
181 SCI 3
182 CF 22
183 END
```

5. SV. This program is a short, root-solving program which uses the secant method to iterate to a real solution of arbitrary, user-defined functions of the form: F(x) = 0. It is included here, since it is used as a subroutine by other programs in this write-all configuration. It may, also, be used for other applications as well.

The SV routine expects to find the name of the global label of the function to be solved in register 6 and an initial guess for X in register X (the display.) Do NOT use an initial guess of zero for X. Registers 6, 7, 8, and 9 are used by SV. XEQing SV will then result in an iterative, hopefully convergent, search for a value of X which makes the function identified in register 6 equal to zero. The successive values of X are displayed, so you can watch the search converge (or diverge). Termination of the search automatically occurs, when two successive values of X are equal within the resolution of the current display setting. This means that the accuracy of the answer may be determined by pre-selecting a display setting. Upon termination, the solution for X is in the X register (display) and in register 7. NOTE: When a program is running, the PRGM annunciator is visible in the display. When the program stops, the annunciator goes out.

For example, suppose we wish to find the solution to the equation:

X = 10^(- X).

This must be rewritten to take the form F(x) = 0, so:

0 = X - 10^(- X).

A program which performs this function and which starts with a global label must be written. Any two-letter, alphabetic label is global, so let's use the label "FN". (Assume X starts in the X register, and the answer is to be left in the X register.) The program would then look like:

LBL FN	The global label so SV can call it.
ENTER	Starting value in X; duplicate it in Y.
10^X	Raise 10 to the X power.
1/X	Take the reciprocal (yields 10^(- X)).
–	Subtract X from Y, leaving answer in X.
RTN	Return to calling routine.

Next, place the label of the function to be solved into register 6: ALPHA FN SHIFT STO 06 ALPHA.
Set the display to 5 significant figures: SHIFT FIX 5
Key in an initial guess: 1
Execute the SV routine: XEQ ALPHA SV ALPHA

The routine will be observed to go through six iterations; finally, stopping at a solution for X of .39901. Executing FN with this value in the X register produces a result of 1 E-10 which is reasonably close to zero.

SV properly converges for routines in this WALL. However, for some user functions, it may have problems with bad initial guesses, or where multiple solutions exist (particularly, when some are imaginary), or when Xs near a root are invalid arguments (e.g.,SQR(X) 0 can't tolerate negative Xs.)

PROGRAM LISTING FOR SV

```
08 AUG 86

    PRP "SV"

01♦LBL "SV"
02 STO 07
03 XEQ IND
         06
04 STO 08
05 RCL 07
06 1
07 %
08 STO 09
09 ST- 07

10♦LBL 00
11 VIEW 07
12 RCL 07
13 XEQ IND
         06
14 ENTER↑
15 ST- 08
16 X<> 08
17 /
18 ST* 09
19 RCL 09
20 ST- 07
21 RCL 07
22 +
23 RND
24 RCL 07
25 RND
26 X≠Y?
27 GTO 00
28 RCL 07
29 RTN
30 .END.
```

6. GR-DIFF.

SHIFT A	SHIFT B	SHIFT C	SHIFT D	SHIFT E
ENTER or FIND wavelength for present condition	ENTER or FIND detector temp for present condition	ENTER or FIND R or RoA for present condition	INTERCHANGE CONDITION 1 AND 2	TOGGLE GR/DIFF
A	B	C	D	E

Below is a table showing six example pieces of data describing two different test conditions. Any five of these may be entered, and the program will solve for the sixth. The solution will assume either a diffusion limited detector or a G-R limited detector, depending on the status of the GR-DIFF toggle.

CONDITION	WAVELENGTH (μM)	TEMPERATURE (K)	R or RoA
1	9	80	10
2 (DIFF LTD.)	11.5	65	4.8
2 (G-R LTD.)	11.5	65	6.9

NOTE: If you can't remember the key definitions, press SHIFT RTN R/S.

The purpose of this program is to permit you to enter reported performance data for detectors which have the "wrong" cutoff or were tested at the "wrong" temperature and to extrapolate how the detectors would perform, if they had the "right" cutoff and/or were tested at the "right" temperature. Or, for example, you can take reported data and predict what temperature and/or wavelength of operation would be required to meet a desired R or RoA. One must assume a measure of quality of the detectors by specifying whether they are diffusion limited or G-R limited. this is done by pressing the GR/DIFF toggle. Current status is indicated prior to any result being displayed.

Only three keys (A, B, and C) are used to enter any of the six data. A is used for wavelength for both test conditions. B is used for temperature for both test conditions. C is used for R or RoA for both test conditions. Alternating between conditions is done with the D key. After any entry, the display will show the status of ALL THREE parameters for the current test condition. Read these values from left to right in the same order as keys A, B, and C. That is, wavelength, temperature, R or RoA. The parameters are separated by a ? mark until a calculation is requested which makes all six data consistent.

For example, viewing the sample data at the beginning of par 6., suppose we were presented with test condition 1 and wanted to know what RoA to expect for a diffusion limited diode under test condition 2. If you're not sure whether the GR-DIFF toggle is set to the diffusion limited case, press the toggle once and see what condition occurs. If it's GR; then, press it again. NOTE: Performing a CLRG (SHIFT TAN) clears all registers and defaults to the diffusion limited case. Press 9, A, 80, B, 10, C, D (to interchange which condition keys A, B, and C affect, and which condition is displayed), 11.5, A, 65, B, C (to solve for RoA for this condition). You may toggle between (and view) the two test conditions, at any time, by pressing D. You may toggle between diffusion limited and G-R limited operation, at any time, by pressing E. NOTE: If the calculation has been performed for one mode of operation (Say, G-R), and the other mode is selected, ? marks are restored to the display to indicate inconsistent data.

The program may also be used when one of the test conditions is diffusion limited, and the other is G-R limited. But you have to get from here to there in two steps, and you must know (or assume) the temperature at which the break from diffusion limited to G-R limited operation occurs. For example, suppose a diode with a 4.2 μM cutoff at 195 K has an Ro of 1E6 ohms. The break from diffusion limited to G-R limited operation for these diodes normally occurs near 130 K. You want to know what Ro to expect at 80 K. Strictly speaking, the cutoff will have changed when the diode is cooled to 80 K, but ignore that fact here. The procedure is as follows: CLRG (clears all registers and puts the machine into diffusion-limited regime, 4.2, A, 195, B, 1E6, C (test condition 1 now contains the 195 K performance), D, 4.2, A, 130, B, C (test condition 2 now contains the Ro at 130 K, diffusion limited, Ro = 6.5 E9), E (to toggle to G-R limited), D (test condition 1 will now represent the 80 K performance) (wavelength is still 4.2, so no need to reenter), 80, B, C (to find the G-R limited Ro at 80 K). The answer is Ro = 2.5 E13. More correct would be to use the correct predicted values for the wavelength at 130 K and 80 K, but solution for such parameters won't be covered until discussion of program HCT (par 7.).

PROGRAM LISTING FOR GR-DIFF

```
01/10/1986

01◆LBL "GR-DIFF"
02 "α,T,RA,SHIFT,GD"
03 PROMPT
04◆LBL D
05 RCL 09
06 X<> 10
07 STO 09
08 RCL 08
09 X<> 11
10 STO 08
11 RCL 07
12 X<> 12
13 STO 07
14◆LBL 01
15 CLA
16 FIX 2
17 ARCL 09
18 ARCL 14
19 FIX 0
20 ARCL 08
21 ARCL 14
22 SCI 1
23 ARCL 07
24 PROMPT
25◆LBL 04
26 1
27 RCL 13
28 X=Y?
29 RTN
30 2
31 X=Y?
32 RTN
33◆LBL E
34 "?"
35 ASTO 14
36 RCL 13
37 1
38 X=Y?
39 2
40 STO 13
41◆LBL 02
42 "DIFFUS,N"
43 RCL 13
44 2
45 X=Y?
46 "G-R"
47 "├ LTD"
48 AVIEW
49 RTN
50◆LBL C
51 "?"
52 ASTO 14
53 STO 07
54 FS?C 22
55 GTO 01
56 XEQ 04
57 XEQ 02
58 14388
59 RCL 13
60 /
61 STO 06
62 RCL 09
63 RCL 08
64 *
65 /
66 RCL 06
67 RCL 10
68 RCL 11
69 *
70 /
71 -
72 E↑X
73 RCL 12
74 *
75 STO 07
76 " "
77 ASTO 14
78 GTO 01
79◆LBL B
80 "?"
81 ASTO 14
82 STO 08
83 FS?C 22
84 GTO 01
85 GTO 00
86◆LBL A
87 "?"
88 ASTO 14
89 STO 09
90 FS?C 22
91 GTO 01
92 SF 00
93◆LBL 00
94 XEQ 04
95 XEQ 02
96 RCL 07
97 RCL 12
98 /
99 LN
100 14388
101 /
102 RCL 13
103 *
104 RCL 10
105 RCL 11
106 *
107 1/X
108 +
109 1/X
110 RCL 09
111 RCL 08
112 FS? 00
113 X<>Y
114 RDN
115 /
116 FS? 00
117 STO 09
118 FC?C 00
119 STO 08
120 " "
121 ASTO 14
122 GTO 01
123 END
```

7. **HCT, X, EG.**



SAMPLE DATA AND RESULTS

	CALCULATION TO FIND X			FOR A NEW	USE X TO FIND:	
CUTOFF	**BANDGAP**	**TEMPERATURE**	**X**	**TEMPERATURE**	**BANDGAP**	**CUTOFF**
4.2		195	.31879	80	.27342	4.543
	0.3	195	.32173	80	.27806	4.459

Registers used: 4 — Cutoff wavelength OR bandgap
5 — Temperature
6 — "X" alpha label for SV program
7 — X, either from direct entry or left by SV program
8 — Used by SV
9 — Used by SV

This program uses the empirical equation for bandgap in terms of temperature and X value developed by Hansen, Schmidt, and Casselman (HSC)[1], which is valid for a wide range of X:

$$Eg = -.302 + 1.93X + (5.35E\text{-}4)T(1-2X) - 0.810X \wedge 2 + 0.832X \wedge 3.$$

It is generally desirable to determine the bandgap or cutoff that a device would have at some temperature from data which are available at some other temperature. The parameter which remains fixed in such calculations is the material composition, X. If X is known, it is stored into the program via label C; labels D and E calculate bandgap or cutoff directly from temperature. Sometimes, however, the data provided does not include X, so the HSC equation is solved iteratively for X as a function of T and (Eg or cutoff) through the use of labels A or B. The value found for X is displayed and saved by the program for use by D and E.

This program solves the HSC equation for X by calling the SV program, described elsewhere (par 5.), and searching for a root of the equation HSC − Eg = 0.

Global labels available from this program are HCT which behaves like local label A and may be used to find X from cutoff and temperature, and EG which calculates Eg from an internally stored value for T and an X in the X register. Of course, these may be called from other programs.

[1]Hansen, Schmidt, and Casselman (HSC), J. Appl. Phys. 53, 7099; 1982.

PROGRAM LISTING FOR HCT, X, EG

08 AUG 86

```
01♦LBL "HCT"
02♦LBL A
03 CF 00
04 GTO 01
05♦LBL B
06 SF 00
07♦LBL 01
08 FIX 9
09 "X"
10 ASTO 06
11 STO 05
12 RDN
13 STO 04
14 CLX
15 ENTER↑
16 .25
17 SF 10
18 XEQ "SV"
19 RTN
20♦LBL E
21 XEQ D
22 1.242
23 /
24 1/X
25 RTN
26♦LBL "X"
27 XEQ 00
28 FC? 00
29 1.242
30 RCL 04
31 FC? 00
32 /
33 -
34 RTN
35♦LBL D
36 STO 05
37 RCL 07
38♦LBL 00
39♦LBL "EG"
40 ENTER↑
41 ENTER↑
42 ENTER↑
43 .832
44 *
45 .81
46 -
47 *
48 1.93
49 +
50 5.35 E-4
51 RCL 05
52 *
53 ENTER↑
54 RDN
55 ST+ X
56 -
57 *
58 +
59 .302
60 -
61 RTN
62♦LBL C
63 STO 07
64 END
```

8. NI, RØA.



Below is a table showing the data required, registers used, sample data, and results.

	TEST DATA	REGISTER USED
MINORITY CARRIER LIFETIME (s)	1E-7	1
MAJORITY CARRIER CONCENTRATION (/cm^2)	5E15	2
COMPOSITION (X) OR	.21200	7
CUTOFF WAVELENGTH (μM)	12	4
DETECTOR TEMPERATURE (K)	80	5
BASE LAYER THICKNESS (μM)	10	3
DETECTOR AREA (cm^2) (only req'd for Isat)	6.45E-6	12

	RESULTS FOR CUTOFF = 12
INTRINSIC CARRIER CONCENTRATION (/cm^2)	3.551E13
PREDICTED RoA (ohm*cm^2)	17.064
SATURATION CURRENT (A)	2.606E-9

This program uses the following equation to solve for RoA. It assumes that the device is diffusion limited, that the diffusion length is much longer than the base layer is thick, and that surface effects are not present. Also, lateral collection effects are not considered, unless one takes the A as the effective carrier collecting area and not the junction area.

$$RoA = \frac{k\,T\,Na\,t}{e^{2}\,d\,Ni^{2}},$$

where e is the charge on an electron;
t is the minority carrier lifetime;
k is Boltzmann's constant;
T is the detector temperature;
d is the base layer thickness;
Na is the majority carrier concentration; and
Ni is the intrinsic carrier concentration.

All these parameters are user input, except the intrinsic carrier concentration which must be solved for from the detector temperature (already input), and the material composition (X). The equation used is from Hansen and Schmidt[2]:

$$Ni = [5.585 - 3.820\,X + (1.753E\text{-}3)T - (1.364E\text{-}3)XT]\,[\,(10^{14})Eg^{.75}T^{1.5}e^{-Eg/2kT}].$$

You have the option of providing the material cutoff wavelength instead of X, in which case, the program calls the HCT program to solve for X, and uses (1.24/cutoff) for Eg. If X is provided instead of the cutoff, the program calls the EG program to find Eg. NI keeps track of whether X or the cutoff has been input (and selected to predominate.) For example, if calculations are done for various device temperatures and X predominates, new values for Eg will be calculated at each temperature. If cutoff predominates, new values of X will be calculated at each temperature. If it is desired that the X value predominate, but only the cutoff is known, run the program once entering the known cutoff. Then, EITHER quickly jot down the X value that is briefly displayed and key it in via SHIFT C prior to any further calculations, OR merely clear flag 2.

The diode diffusion limited saturation current may be found, once the RoA is known, IF the diode area is known. The equation used is:

$$Isat = \frac{k\,T}{e\,Ro}, \text{ where } Ro = \frac{Ro\,A}{A}.$$

As a point of interest, viewing the program code for LBL C (Isat), one observes that, despite the fact that RØA should be executed as a subroutine, the RØA routine is not run via an XEQ (which is a subroutine call) but via a GTO (which is a jump); and that RØA returns to the Isat routine based on the status of flag 3. The reason for this more-involved method for calling RØA lies in the fact that the HP41's RTN (return) stack is, only, six levels deep. This stack is where the HP41 keeps track of the location from which a subroutine has been called (the location where it must return at the end of a subroutine.) Following the program code, one finds that RØA calls NI, which may call label 01, which calls HCT, which calls SV, which calls X, which calls EG. If one tries to use RØA as a subroutine, the return address in the initial calling program gets pushed off the stack (and lost), when EG gets called; so the program terminates (incorrectly) in RØA.

[2]Hansen and Schmidt, J. Appl. Phys. 54, 3; March 1983.

PROGRAM LISTING FOR NI, RØA

```
01/10/1986

01♦LBL "NI"
02♦LBL E
03 RCL 11
04 FS? 01
05 RTN
06 FS? 02
07 XEQ 01
08 FC? 02
09 XEQ 02
10 -1.364
11 RCL 05
12 *
13 3820
14 -
15 *
16 1.753
17 RCL 05
18 *
19 +
20 5585
21 +
22  E11
23 *
24 1.24
25 RCL 04
26 /
27 SQRT
28 RCL 05
29 *
30 SQRT
31 3
32 Y↑X
33 *
34 7194
35 RCL 04
36 /
37 RCL 05
38 /
39 E↑X
40 /
41 SF 01
42 STO 11
43 CLD
44 FIX 3
45 RTN
46♦LBL 01
47 RCL 04
48 RCL 05
49 XEQ "HCT"
50 RTN
51♦LBL 02
52 RCL 07
53 XEQ "EG"
54 1.24
55 /
56 1/X
57 FIX 2
58 "α="
59 ARCL X
60 AVIEW
61 STO 04
62 RCL 07
63 RTN
64♦LBL "R0A"
65♦LBL D
66 FIX 3
67 XEQ E
68 5378 E15
69 X<>Y
70 X↑2
71 /
72 RCL 05
73 *
74 RCL 02
75 *
76 RCL 03
77 /
78 RCL 01
79 *
80 "RA="
81 ARCL X
82 AVIEW
83 FS?C 03
84 GTO 03
85 RTN
86♦LBL B
87 STO 12
88 RTN
89♦LBL C
90 SF 03
91 GTO D
92♦LBL 03
93 RCL 12
94 /
95 RCL 05
96 /
97 11605
98 *
99 1/X
100 "I="
101 ARCL X
102 PROMPT
103♦LBL a
104 STO 01
105 RTN
106♦LBL b
107 STO 02
108 RTN
109♦LBL c
110 STO 07
111 CF 01
112 CF 02
113 RTN
114♦LBL d
115 STO 04
116 CF 01
117 SF 02
118 RTN
119♦LBL e
120 STO 05
121 CF 01
122 RTN
123♦LBL A
124 STO 03
125 END
```

9. CONSTNT.

c speed of light	e_0 permitivity of free space	sigma Stefan- Boltzmann constant	R gas constant	N_A Avogadro's number
SHIFT A	SHIFT B	SHIFT C	SHIFT D	SHIFT E
k Boltzmann constant	e charge on an electron	h Plank constant	h/(2*PI) Plank constant divided by (2*PI)	m_e mass of an electron
A	B	C	D	E

This program may be used to conveniently bring any of the available physical constants into the X register for use in direct keyboard calculations. All constants are in MKS units. When any of the keys are pressed (SHIFT A through SHIFT E, or A through E) the current value in the X register is pushed up into the Y register, and the selected constant is placed in the X register. For example, suppose one wants to evaluate the following expression:

$$\frac{h\ c}{\text{wavelength}\ k\ T},$$

where wavelength = 4.2E-6 m,
and T = 300 K.

The sequence that may be followed is:

PRESS	TO OBTAIN	SEE
SHIFT SCI 4	Floating point display and 5 significant figures	
C	h	6.6262 -34
SHIFT A	c	2.9979 08
*	Multiply X and Y registers	1.9865 -25
4.2 EEX CHS 6	Wavelength	4.2 -6
/	Divide Y register by X	4.7297 -20
A	k	1.3806 -23
/	Divide Y register by X	3.4258 03
300	T	300
/	Divide Y register by X	1.1419 01

PROGRAM LISTING FOR CONSTNT

```
01/10/1986

♦LBL "CONSTNT"
♦LBL a
2.997925 E8
RTN
♦LBL A
1.38062 E-23
RTN
♦LBL B
1.6022 E-19
RTN
♦LBL C
6.6262 E-34
RTN
♦LBL D
1.0546 E-34
RTN
♦LBL E
9.1096 E-31
RTN
♦LBL e
6.02217 E23
RTN
♦LBL b
8.85 E-12
RTN
♦LBL c
5.66961 E-8
RTN
♦LBL d
8.31434
RTN
END
```

10. INJEFF.

detector area (sq. cm.)	wavelength cuton (μM) ENTER cutoff (μM)	background temperature (K)	quantum efficiency (percent)	F/# (cold shield)
SHIFT A	SHIFT B	SHIFT C	SHIFT D	SHIFT E
zero bias resistance (ohms)	detector bias (volts)	detector temperature (K)	detector m value (1 = diff. ltd) ENTER CCD input derating factor (usually = 1.4)	FIND Rd Q Ip Iv Rf IE
A	B	C	D	E

Below is a table showing the data required, registers used, sample data, and results:

PARAMETER	REGISTER USED	SAMPLE DATA	RESULTS
DETECTOR AREA, A	16	1E-5	
CUTON	5	3.0	
CUTOFF	10	5.0	
BACKGROUND TEMPERATURE	12	300	
QUANTUM EFFICIENCY	17	80	
COLD SHIELD F/#	13	1	
ZERO BIAS RESISTANCE, Ro	9	1E6	
DETECTOR BIAS, V	2	- .02	
DETECTOR TEMPERATURE	14	200	
DETECTOR m VALUE (1 = DIFF. LTD., 2 = G-R LTD.)	3	1.2	
CCD DERATING FACTOR (m FOR CCD INPUT NODE)	11	1.4	
Rd = DIODE DYNAMIC RESISTANCE (OHMS)	1		2.63E6
Q = PHOTON FLUX (/cm²/s)	15		1.31E16
Ip = DIODE PHOTOCURRENT (A)			4.20E-9
Iv = DIODE BIAS CURRENT (A)			1.28E-8
Rf = CCD INPUT IMPEDANCE (OHMS) (RECIPROCAL OF TRANSCONDUCTANCE)			1.42E6
IE = INJECTION EFFICIENCY			0.650

Other registers used: 4, 6, 7, 8

This program calculates injection efficiency from the equation:

$$IE = \frac{Rd}{(\text{CCD derating factor})Rf + Rd}.$$

Rf is the CCD input impedance and is the reciprocal of the term normally used: CCD input transconductance. Rf is computed from:

$$Rf = \frac{kT}{e\,Id}, \quad \text{where Id is the net diode current} = Iv + Ip.$$

Ip is the photocurrent induced in the photodiode by the background photon flux, Q, and is:

$$Ip = e\,A\,Q\,(\text{quantum efficiency}).$$

Iv is the diode current due to the fact that a bias is applied and is found from the equation of a diode:

$$Iv = \frac{mkT\,[EXP(-eV/\{mkT\}) - 1]}{e\,Ro}.$$

Finally, Rd is the dynamic resistance of the diode at the specified bias. This may be found by differentiating the equation for Iv with respect to V and taking the reciprocal, so that:

$$Rd = Ro\,EXP(-eV/\{mkT\}).$$

When E is pressed to compute an injection efficiency, the intermediate results (Rd, Q, Ip, Iv, Rf), with identification, are flashed in the display, as they are found. The program stops with the injection efficiency displayed. If a printer is connected to the HP 41 and is on before the HP41 is turned on, all results will be printed. These data flash past too fast to be recorded without a printer and, if you're not interested in it, just ignore it. If your ARE interested in the intermediate results, set flag 21 using SHIFT SF 21 prior to pressing E. The HP 41 will stop at each intermediate value so that it may be recorded. Pressing R/S at each of these points will continue the program to the next result, ultimately displaying the value for the injection efficiency. You need set flag 21 only once, and it will remain set until you clear it using the FLAG routine, using SHIFT CF 21, or by turning the HP41 on with no printer connected.

PROGRAM LISTING FOR INJEFF

```
01/10/1986

01♦LBL "INJEFF"
02♦LBL a
03 STO 16
04 RTN
05♦LBL b
06 STO 10
07 X<>Y
08 STO 05
09 CF 00
10 RTN
11♦LBL c
12 STO 12
13 CF 00
14 RTN
15♦LBL d
16 STO 17
17 RTN
18♦LBL e
19 STO 13
20 RTN
21♦LBL A
22 STO 09
23 RTN
24♦LBL B
25 STO 02
26 RTN
27♦LBL C
28 STO 14
29 RTN
30♦LBL D
31 STO 11
32 X<>Y
33 STO 03
34 RTN
35♦LBL E
36 SCI 2
37 11605
38 RCL 14
39 /
40 STO 06
41 RCL 02
42 *
43 RCL 03
44 /
45 E↑X
46 STO 04
47 1/X
48 RCL 09
49 *
50 STO 01
51 "Rd="
52 ARCL X
53 AVIEW
54 FS? 00
55 GTO 00
56 CF 02
57 RCL 10
58 XEQ "QBB"
59 STO 15
60 RCL 05
61 XEQ "QBB"
62 ST- 15
63 SF 00
64♦LBL 00
65 RCL 15
66 "Q="
67 ARCL X
68 AVIEW
69 RCL 17
70 *
71 RCL 13
72 X↑2
73 /
74 4 E-22
75 *
76 RCL 16
77 *
78 CHS
79 "IP="
80 ARCL X
81 AVIEW
82 RCL 04
83 1
84 -
85 RCL 06
86 /
87 RCL 09
88 /
89 RCL 03
90 *
91 "IV="
92 ARCL X
93 AVIEW
94 +
95 ABS
96 RCL 06
97 *
98 1/X
99 RCL 11
100 *
101 "RF="
102 ARCL X
103 AVIEW
104 RCL 01
105 +
106 RCL 01
107 /
108 1/X
109 "IE="
110 FIX 3
111 ARCL X
112 AVIEW
113 END
```

III. FINAL COMMENTS

As with any software, the results are only as valid as the data, the equations, and the program steps which are intended to represent them. The various programs have been tested and are believed to be correct and functional. However, if any errors are found, if any problems arise, or if you have and comments or questions, please let the author of this report know.

DISTRIBUTION FOR REPORT NO. AMSEL-NV-TR -0057

Number of Copies	Addressee
1	Commander AMMRC Watertown, MA 02172
1	Commandant US Army Infantry School Ft. Benning, GA 31905
1	Director Defense Advanced Research Projects Agency 1400 Wilson Blvd. Rosslyn, VA 22209
1	Commander US Naval Research Lab Washington, DC 20375
1	Commander HQ DARCOM ATTN: DRCCP-E Alexandria, VA 22333
1	Commandant Defense Systems Management College ATTN: DRI-R Fort Belvoir, VA 22060-5426
1	Commandant US Army Engineer School Fort Belvoir, VA 22060
1	Commander US Operational Test and Evaluation Agency ATTN: Tech Library 5600 Columbia Pike, Room 503 Falls Church, VA 22041
1	Commander USACSC Fort Belvoir, VA 22060
1	HQDA ATTN: DACA-CA Washington, DC 20310
2	Defense Technical Information Center ATTN: DTIC-FDA Cameron Station, Bldg 5 Alexandria, VA 22304-6145
1	Commander MIRADCOM ATTN: DRDMI-DC Redstone Arsenal, AL 35809
1	Commander MIRCOM ATTN: DRSMI-FO Redstone Arsenal, AL 35809
1	Commander AVARADCOM ATTN: DRCAV-BC St. Louis, MO 63166
1	Commander TSARCOM ATTN: DRSTS-CO St. Louis, MO 63166
1	Commander TARADCOM ATTN: DRCTA-VC Warren, MI 48090
1	Commander US Army Tank Automotive Command ATTN: DRSTA-TSL Warren, MI 48090
1	Commander TARCOM ATTN: DRSTA-NC Warren, MI 48090
1	Commander NARADCOM ATTN: DRDNA-O Natick, MA 01760
1	Commander TRADOC ATTN: ATCD-AN Fort Monroe, VA 23651

Number of Copies	Addressee
2	Commander US Army Logistics Ctr Fort Lee, VA 23801
1	Commander US Army Systems Analysis Agency Aberdeen Proving Ground, MD 21005
1	NASA Scientific & Tech Info Facility ATTN: Acquisitions Branch (S-AK/DL) Box 8757 BWI Airport, MD 21040
1	Study Center National Maritime Research Ctr King's Point, NY 11024
1	Commander USAAVNC Ft. Rucker, AL 36362
20	Commander M. Geisler ERADCOM ATTN: DRDEL-AP-OA Adelphi, MD 20783
1	Director Atmospheric Sciences Lab ATTN: DELAS-D White Sands Missile Range, NM 88002
1	Director CS&TA Laboratory ATTN: DELCS-D Fort Monmouth, NJ 07703
1	Director Electronic Warfare Lab ATTN: DELEW-D Ft. Monmouth, NJ 07703
1	Director Electronics Technology & Devices Lab ATTN: DELET-D Ft. Monmouth, NJ 07703
1	Commander Harry Diamond Labs ATTN: DELHD-AC Adelphi, MD 20783

Number of Copies	Addressee
1	Director Signal Warfare Lab ATTN: DELSW-D Vint Hill Station, VA 22186
1	Project Manager XM-1 Tank System Warren, MI 48090-5000
1	Project Manager M60 Tank System Warren, MI 48090-5000
1	Project Manager MICV Warren, MI 48090-5000
1	Project Manager TOW/DRAGON Redstone Arsenal, AL 35809
1	Project Manager GLID/LTD Redstone Arsenal, AL 35809
1	Project Manager VIPER/AHAMS Redstone Arsenal, AL 35809
1	Project Manager AAH AVARADCOM St. Louis, MO 63166
1	Project Manager TADS/PNVS AVARADCOM St. Louis, MO 63166
1	Commander ARRCOM ATTN: DRSAR-CPE Rock Island, IL 61299
1	Project Manager RPV AVARADCOM St. Louis, MO 63166

Number of Copies	Addressee
1	Commander ARRADCOM ATTN: DRDAR-SEC Dover, NJ 07801
1	Commander CORADCOM ATTN: DRDCO-PPA-CA Fort Monmouth, NJ 07703
1	Commander CERCOM ATTN: DRSEL-CP-CR Fort Monmouth, NJ 07703
1	Commander Belvoir RD&E Ctr ATTN: STRBE-BT
3	ATTN: STRBE-BPG Fort Belvoir, VA 22060-5606
1	Director CNVEO ATTN: DELNU-TMS/SEMCO Ft. Belvoir, VA 22060
1	Project Manager REMBASS ATTN: DRCPM-RBS Fort Monmouth, NJ 07703
1	Project Manager FIREFINDER ATTN: DRCPM-FF Fort Monmouth, NJ 07703
1	Department of Defense Production Engineering Spt Ofc (PESO) ATTN: D. Anderson Cameron Station Alexandria, VA 22314
1	Project Manager CAC ATTN: DRCPM-CAC Vint Hill Station, VA 22186

Number of Copies	Addressee
1	Project Manager SOTAS ATTN: DRCPM-STA Fort Monmouth, NJ 07703
1	Commandant US Army Air Defense School Fort Bliss, TX 79916
1	Commandant US Army Armor School Fort Knox, KY 40121
1	Commandant US Army Field Artillery School Fort Sill, OK 73503
1	Director US Army Research and Technology Labs Ames Research Ctr Moffett Field, CA 94035
3	Commander AFSC/ACC Andrews AFC, MD 20334
2	Commander ESD/ACC Hanscomb AFB, MA 01730
1	Commander US Naval Surface Weapons Ctr/White Oak ATTN: Technical Library Silver Spring, MD 20910
1	Commander Naval Electronics Lab Ctr Library San Diego, CA 92152
1	Armament Development and Test ATTN: DLOSL Tech Library Elgin Air Force Base, FL 32542

Number of Copies	Addressee
1	Honeywell, Inc. Electro-Optics Operations Defense Electronics Division ATTN: Bill Lepsevich 2 Forbes Road Lexington, MA 02173
1	Honeywell, Inc. Electro-Optics Operations Focal Plane Array Laboratory ATTN: Shigesato Iwasa 2 Forbest Road Lexington, MA 02173
1	Honeywell, Inc. Systems & Research Center ATTN: Norman A. Foss (MM17-2334) 2600 Ridgway Parkway Minneapolis, MN 55413
1	HRB, Inc. P.O. Box 60 State College, PA 16801
1	Hughes Aircraft Company Centinela & Teale Streets Culver City, CA 90230
1	Hughes Aircraft Company Research Laboratories 3011 Malibu Canyon Road Malibu, CA 90265
1	Hughes Missile Systems Div. ATTN: Herm Latt 8433 Fallbrook Avenue Canoga Park, CA 91304
1	Institute of Defense Analyses ATTN: W. Holzer 1801 N. Beauregard Street Alexandria, VA 22311
1	ITT Electro-Optical Products Division 7635 Plantation Road Attn: Mr. R. Slusher Roanoke, VA 24019

Number of Copies	Addressee
1	Litton Electron Tube Division ATTN: Mr. Arnold Davis 1215 S. 52nd Street Tempe, AZ 85281
1	Magnavox Government & Industrial Electronics Corp. Electro-Optical Systems ATTN: N. Diepeveen 46 Industrial Avenue Mahwah, NJ 07430
1	Magnavox Government & Industrial Electronics Corp. Electro-Optical Systems ATTN: Mr. Charles Arduini 46 Industrial Avenue Mahwah, NJ 07430
1	Command, Control & Communications Div. Development Center Marine Corps Development & Educ. Command Quantico, VA 22134
1	Marlow Industries Incorporated 1021 S. Jupiter Road ATTN: Mr. R. Marlow Garland, TX 75042
1	Martin Marietta Aerospace Orlando Div. P.O. Box 5837 ATTN: Dr. R. Landrum M/S 362 Orlando, FL 32885
1	Optoelectronics Inc. 1309 Dynamic Street Petaluma, CA 94952
1	Perkin-Elmer Corp. Main Avenue Norwalk, CT 06852
1	Commander Picatinny Arsenal ATTN: SARPA-TS-S #59 Dover, NJ 07801

Number of Copies	Addressee
1	Raytheon Company Special Microwave Devices Operation Bear Foot Road Northborough, MA 01532
1	RCA Corporation David Sarnoff Research Center P.O. Box 432 Princeton, NJ 08540
1	RCA Corporation Government Systems Division ATTN: Mike Cantella P.O. Box 588 Burlington, MA 01803
1	Rockwell International Corp. ATTN: Dr. Mike Andrews P.O. Box 4192 3370 Miraloma Ave Anaheim, CA 92803
1	Rockwell International Corp. Science Center ATTN: Jon Rode 1049 Camino Dos Rios Thousand Oaks, CA 91460
1	Santa Barbara Research Center ATTN: James Scorso 75 Coromar Drive Coleta, CA 93117
1	McDonnell Douglas Astronautics Company ATTN: Mr. J. Kline ?301 Bolsa Avenue Huntington Beach, CA 92647
1	M.I.T Lincoln Laboratory P.O. Box 73 Lexington, MA 02173
1	Commander Naval Ocean Systems Center ATTN: Library San Diego, CA 92152

Number of Copies	Addressee
1	Director Naval Research Laboratory ATTN: J.M. Killiany Code 6813 Washington, DC 20375
1	Commander Naval Surface Weapons Center ATTN: Technical Library White Oak, Silver Spring, MD 20910
1	New England Research Center ATTN: R. Rotolante Minuteman Dr./Longfellow Ct. Sudbury, MA 01776
1	Ni-Tec Incorporated 7426 Linder Avenue ATTN: Mr. W. Woods Skokie, IL 60076
1	Northrop Corp. Research & Technology Ctr. ATTN: Dean Baker One Research Park Palos Verdes Peninsula, CA 90274
1	Northrop Corp. Electro-Mechanical Div. ATTN: R. Helfrich M/S Y32/7301 500 East Orangethrope Ave. Anaheim, CA 92801
1	Office of Naval Research Code 427 Arlington, VA 22217
1	Commander HQ Master Technical Information Center ATTN: Mrs. R. Reynolds Fort Hood, TX 76544
1	Texas Instruments, Inc. ATTN: Mr. W. Irby P.O. Box 22601 Dallas, TX 75265

Number of Copies	Addressee
1	Texas Instruments, Inc. Central Research Laboratories ATTN: C. Grady Roberts M/S 132 P.O. Box 225936 Dallas, TX 75265
1	USA Security Agency ATTN: IARD Arlington Hall Station Arlington, VA 22212
1	Commander U.S. Army Air Defense School ATTN: ATSA-TSM-S (Col. Leach) Ft. Bliss, TX 79916
1	Headquarters U.S. Army Air Defense School ATTN: ATSA-CDM-W (Cpt. L. Bourneuf) Ft. Bliss, TX 79916
1	Commander U.S. Army Armament Command ATTN: AMSAR-RDP (Library) Rock Island, IL 61201
1	Commander U.S. Army Infantry School ATTN: ATSH-CD-MS-E (Mr. R. McKenna) Ft. Benning, GA 31705
1	Commandant U.S. Army Institute for Military Assistance ATTN: ATSU-CTD-OMS Fort Bragg, NC 28307
1	Commander U.S. Army Logistics Center ATTN: ATCL-MC Fort Lee, VA 22801
1	Commander U.S. Army Missile Command ATTN: DRCPM-MPE (Mr. R. Dudney) Redstone Arsenal, AL 35898
1	Command General U.S. Army Missile Command ATTN: DRSMI-QRT (G. Hutcheson) Redstone Arsenal, AL 35809
1	Commander U.S. Army Missile Command ATTN: AMSMI-RE (Mr. Pittman) Redstone Arsenal, AL 35809
1	Commander U.S. Army Missile Command ATTN: AMSMI-RR, Dr. J.P. Hallowes Redstone Arsenal, AL 35809
1	Commander U.S. Army Missile Command ATTN: DRSMI-XB (Mr. M. Hartwell) Redstone Arsenal, AL 35809
1	U.S. Army Research Office-Durham ATTN: GRDARD-IP Box CM, Duke Station Durham, NC 27706
1	U.S. Army Research Office-Durham ATTN: Dr. Robert J. Lontz Box CM, Duke Station Durham, NC 27706
1	Commander U.S. Army Systems Analysis Agency ATTN: AMXSY-T (Mr. A. Reid) Aberdeen Proving Ground, MD 21005
1	Varian Associates, Inc. Light Sensing & Emitting Division ATTN: Mr. Dominic Ruggieri 611 Hansen Way Palo Alto, CA 94303
1	Varo, Inc., Texas Division P.O. Box 401146 ATTN: Mr. Charles Hanson 2201 Walnut Street Garland, TX 75040
1	Varo, Inc., Texas Division ATTN: Mr. William Hart 2001 Walnut Street Garland, TX 75040

Number of Copies	Addressee	Number of Copies	Addressee
50	Director CNVEO ATTN: David R. Kaplan AMSEL-RD-NV-IT Fort Belvoir, VA 22060-5677		
1	Commander Sacramento Army Depot Sacramento, CA 95813-5009		
1	Commander New Cumberland Army Depot New Cumberland, PA 17070-5001		
1	Commander Anniston Army Depot Anniston, AL 36201		
1	Dept of Defense Production Engineering Spt Ofc (PESO) ATTN: H.K. MacKechnie Cameron Station Alexandria, VA 22314		
1	General Atomic Company P.O. Box 81608 ATTN: Mr. H.J. Snyder, Jr. San Diego, CA 92138		
1	General Dynamics Pomona Division P.O. Box 2507 ATTN: W.J. Baldwin Pomona, CA 91766		
1	Honeywell, Inc. Defense Systems Division ATTN: Mr. James E. Nosan 600 2nd Street NE Hopkins, MN 55343		